*------------- Upper-Body Strength and Political Egalitarianism: Twelve Conceptual Replications ------------------------------------*
*----------------------------------------------------- STUDY 1 ---------------------------------------------------------------------*

*** Background variables 

* Sex
clonevar sex = gender

* Age
summ age

* Education
tab profile_education

clonevar education = profile_education

** Demographic overview:
tab sex
summ age
tab education


*** Dependent variables
* Social Dominance Orientation (larger value = higher in SDO)
alpha q13_1 q13_2 q13_3 q13_4 q13_5 q13_6 q13_7 q13_8

pwcorr q13_1 q13_2 q13_3 q13_4 q13_5 q13_6 q13_7 q13_8, sig
recode q13_3 (0=6) (1=5) (2=4) (3=3) (4=2) (5=1) (6=0)
recode q13_4 (0=6) (1=5) (2=4) (3=3) (4=2) (5=1) (6=0)
recode q13_7 (0=6) (1=5) (2=4) (3=3) (4=2) (5=1) (6=0)
recode q13_8 (0=6) (1=5) (2=4) (3=3) (4=2) (5=1) (6=0)
alpha q13_1 q13_2 q13_3 q13_4 q13_5 q13_6 q13_7 q13_8
egen SDO = rowmean(q13_1 q13_2 q13_3 q13_4 q13_5 q13_6 q13_7 q13_8)

* Recodes to a 0-1 scale for observed min and max for males and females, respectively
summ SDO if sex == 1
generate sdo = (SDO-(r(min)))/(r(max)-r(min)) if sex == 1

summ SDO if sex == 2
replace sdo = (SDO-(r(min)))/(r(max)-r(min)) if sex == 2

bys sex: summ sdo


*** Sub-Dimensions of SDO (again coded to 0-1 scales reflecting in and max observed values within sex):
* SDO-D
egen SDO_D = rowmean(q13_1 q13_2 q13_3 q13_4)
summ SDO_D if sex == 1
generate sdo_d = (SDO_D-(r(min)))/(r(max)-r(min)) if sex == 1

summ SDO_D if sex == 2
replace sdo_d = (SDO_D-(r(min)))/(r(max)-r(min)) if sex == 2

bys sex: summ sdo_d

* SDO-E
egen SDO_E = rowmean(q13_5 q13_6 q13_7 q13_8)
summ SDO_E if sex == 1
generate sdo_e = (SDO_E-(r(min)))/(r(max)-r(min)) if sex == 1

summ SDO_E if sex == 2
replace sdo_e = (SDO_E-(r(min)))/(r(max)-r(min)) if sex == 2

summ sdo_e


*** Authoritarian values
alpha q12_1 q12_2 q12_3 q12_4 q12_5 q12_6
pwcorr q12_1 q12_2 q12_3 q12_4 q12_5 q12_6, sig
recode q12_1 (0=6) (1=5) (2=4) (3=3) (4=2) (5=1) (6=0)
recode q12_3 (0=6) (1=5) (2=4) (3=3) (4=2) (5=1) (6=0)
recode q12_6 (0=6) (1=5) (2=4) (3=3) (4=2) (5=1) (6=0)
alpha q12_1 q12_2 q12_3 q12_4 q12_5 q12_6

egen AUTHO = rowmean(q12_1 q12_2 q12_3 q12_4 q12_5 q12_6)

* Recoded to a 0-1 scale for observed min and max for amles and females, respectively:
summ AUTHO if sex == 1
generate autho = (AUTHO-(r(min)))/(r(max)-r(min)) if sex == 1

summ AUTHO if sex == 2
replace autho = (AUTHO-(r(min)))/(r(max)-r(min)) if sex == 2

summ autho


*** Independent variables
* Self-perceived physical strength
generate strength = (q11-1)/10
bys sex: summ strength


*** Potential moderator of relationship between physical strength and anti-egalitarianism
* Self-perceived social staus (McArthur ladder)
generate status_ladder = (q10-1)/9


*----------------------------------------------- Descriptives ----------------------------------------------------*
bys sex: summ strength

alpha q13_1 q13_2 q13_3 q13_4 q13_5 q13_6 q13_7 q13_8
bys sex: summ sdo

alpha q12_1 q12_2 q12_3 q12_4 q12_5 q12_6
bys sex: summ autho



*------------------------------------------------ MAIN ANALYSES ---------------------------------------------------*
*** Predicting SDO from strength for males and females, respectively:
bys sex: reg sdo c.strength c.age c.education

* Predicting SDO-D & SDO-E from strength for males and females, respectively:
bys sex: reg sdo_d c.strength c.age c.education
bys sex: reg sdo_e c.strength c.age c.education

* Predicting Authoritarianism from strength for males and females, respectively:
bys sex: reg autho c.strength c.age c.education


*** Possible interactions between strength and self-perceived social status in predicting SDO:
* Female subjects
reg sdo c.strength##c.status_ladder c.age c.education if sex == 1

margins, dydx(strength) at(status_ladder=(0 (0.05) 1.0))
marginsplot, recastci(rline) ciopts(lpattern(dash)) recast(line) yline(0) xlabel(#10) ///
xtitle(Socio-Economic Status) ytitle(Marg. effect of self-perceived formidality) title("") scheme(s2mono)

* Male subjects
reg sdo c.strength##c.status_ladder c.age c.education if sex == 2

margins, dydx(strength) at(status_ladder=(0 (0.05) 1.0))
marginsplot, recastci(rline) ciopts(lpattern(dash)) recast(line) yline(0) xlabel(#10) ///
xtitle(Socio-Economic Status) ytitle(Marg. effect of self-perceived formidality ) title("") scheme(s2mono)

